

14 Mplus inputs

14.1 Mplus input for simulation of continuous M and continuous Y with treatment-mediator interaction

[Table 25 about here.]

[Table 26 about here.]

14.2 Mplus input for simulation of binary outcome and continuous mediator

[Table 27 about here.]

[Table 28 about here.]

[Table 29 about here.]

[Table 30 about here.]

14.3 Mplus input for aggression example

[Table 31 about here.]

[Table 32 about here.]

[Table 33 about here.]

14.4 Mplus input for smoking example

[Table 34 about here.]

[Table 35 about here.]

14.5 Mplus input for Pearl's artificial example

[Table 36 about here.]

[Table 37 about here.]

[Table 38 about here.]

[Table 39 about here.]

[Table 40 about here.]

[Table 41 about here.]

[Table 42 about here.]

[Table 43 about here.]

14.6 Mplus input for n=200 data drawn on the Pearl example

[Table 44 about here.]

[Table 45 about here.]

14.7 Mplus input for Monte Carlo simulation of a nominal mediator and a continuous outcome

[Table 46 about here.]

[Table 47 about here.]

[Table 48 about here.]

[Table 49 about here.]

14.8 Mplus input for hypothetical pollution data with a nominal mediator and a binary outcome

[Table 50 about here.]

[Table 51 about here.]

14.9 Mplus input for count outcome simulation

[Table 52 about here.]

[Table 53 about here.]

[Table 54 about here.]

14.10 Mplus input for sensitivity analysis

14.10.1 Regular mediation with $\rho = 0$

[Table 55 about here.]

14.10.2 Sensitivity analysis with $\rho = 0.25$

[Table 56 about here.]

14.10.3 Analysis of head circumference

[Table 57 about here.]

Table 25: Input for step 1 y on xm

```

TITLE:      Simulating x-m interaction effect on y using a random
            slope, saving the data for external Monte Carlo analysis
MONTECARLO:
            NAMES = y m x;
            NOBSEVATION = 400;
            NREPS = 500;
            REPSAVE = ALL;
            SAVE = xmrep*.dat;
            CUTPOINTS = x(0);
MODEL POPULATION:
            x@1;
            [y*1];
            y ON x*.4;
            beta1 | y ON m;
            beta1 ON x*.2;
            [beta1*.5];
            beta1@0;
            [m*2];
            m ON x*.5;
            y*.5;
            m*1;
ANALYSIS:  TYPE = RANDOM;
MODEL:     [y*1] (beta0);
            y ON x*.4 (beta2);
            beta1 | y ON m;
            beta1 ON x*.2 (beta3);
            [beta1*.5] (beta1);
            beta1@0;
            [m*2] (gamma0);
            m ON x*.5 (gamma1);
            y*.5;
            m*1;
MODEL CONSTRAINT:
            NEW(tie*.35 pie*.25 de*.8);
            tie=beta1*gamma1+beta3*gamma1;
            pie=beta1*gamma1;
            de=beta2+beta3*gamma0;

```

Table 26: Input for step 2 y on xm

```

TITLE:      Simulating x-m interaction effect on y using a random
            slope, Step 2: External Monte Carlo analysis
DATA:      FILE = xmreplist.dat;
            TYPE = MONTECARLO;
VARIABLE:  NAMES = y m x;
            USEVARIABLES = y m x xm;
DEFINE:    xm = x*m;
ANALYSIS:
MODEL:     [y*1] (beta0);
            y ON x*.4 (beta2);
            y ON xm (beta3);
            y ON m*.5 (beta1);
            [m*2] (gamma0);
            m ON x*.5 (gamma1);
            y*.5;
            m*1;
MODEL CONSTRAINT:
            NEW(tie*.35 pie*.25 de*.8);
            tie=beta1*gamma1+beta3*gamma1;
            pie=beta1*gamma1;
            de=beta2+beta3*gamma0;

```

Table 27: Input for step 1 ML y on xm n=200

```

TITLE:      Simulating x-m interaction effect on y using a random
             slope, saving the data for external Monte Carlo analysis:
             Binary Y, continuous M
MONTECARLO:
             NAMES = y m x;
             GENERATE = y(1 p);
             CATEGORICAL = y;
             NOBSERVATION = 200;
             NREPS = 500;
             REPSAVE = ALL;
             SAVE = n200xmrep*.dat;
             CUTPOINTS = x(0);
MODEL POPULATION:
             x@1;
             [y$1*.5];
             y ON x*.3;
             beta1 | y ON m;
             beta1 ON x*.2;
             [beta1*.7];
             beta1@0;
             [m*.5];
             m ON x*.5;
             m*.75;
ANALYSIS:  TYPE = RANDOM;
             ESTIMATOR = ML;
             LINK = PROBIT;
MODEL:     [y$1*.5] (beta0);
             y ON x*.3 (beta2);
             beta1 | y ON m;
             beta1 ON x*.2 (beta3);
             [beta1*.7] (beta1);
             beta1@0;
             [m*.5] (gamma0);
             m ON x*.5 (gamma1);
             m*.75;
MODEL CONSTRAINT:
             NEW(ind*.45 dir*.4);
             ind=beta1*gamma1+beta3*gamma1;
             dir=beta3*gamma0+beta2;

```

Table 28: Input for step 2 ML y on xm n=200

```

TITLE:      Simulating x-m interaction effect on y using a random slope
            Step 2
            Binary Y, continuous M
DATA:      FILE = n200xmreplist.dat;
            TYPE = MONTECARLO;
VARIABLE:  NAMES = y m x;
            USEVARIABLES = y m x xm;
            CATEGORICAL = y;
DEFINE:    xm = x*m;
ANALYSIS:  ESTIMATOR = ML;
            LINK = PROBIT;
MODEL:     [y$1*.5] (mbeta0);
            y ON x*.3 (mbeta2);
            y ON m*.7 (beta1);
            y ON xm*.2 (beta3);
            [m*.5] (gamma0);
            m ON x*.5 (gamma1);
            m*.75 (sig2);
MODEL CONSTRAINT:
            NEW(ind*.45 dir*.4 arg11*.7 arg10*.25 arg01*.2 arg00*-
            .15 v1*1.6075 v0*1.3675 probit11*.5521 probit10*.1972
            probit01*.17103 probit00*-.1283 tie*.131 de*.129
            pie*.119);
            dir=beta3*gamma0+beta2;
            ind=beta1*gamma1+beta3*gamma1;
            arg11=-mbeta0+beta2+(beta1+beta3)*(gamma0+gamma1);
            arg10=-mbeta0+beta2+(beta1+beta3)*gamma0;
            arg01=-mbeta0+beta1*(gamma0+gamma1);
            arg00=-mbeta0+beta1*gamma0;
            v1=(beta1+beta3)^2*sig2+1;
            v0=beta1^2*sig2+1;
            probit11=arg11/sqrt(v1);
            probit10=arg10/sqrt(v1);
            probit01=arg01/sqrt(v0);
            probit00=arg00/sqrt(v0);
            ! Phi function needed below:
            tie=phi(probit11)-phi(probit10);
            de=phi(probit10)-phi(probit00);
            pie=phi(probit01)-phi(probit00);

```

Table 29: Input excerpts for step 2 bayes y on xm n=200

```

ANALYSIS: ESTIMATOR = BAYES;
           FBITER = 10000;
MODEL:    [y$1*.5] (mbeta0);
           y ON x*.3 (beta2);
           y ON m*.7 (beta1);
           y ON xm*.2 (beta3);
           [m*.5] (gamma0);
           m ON x*.5 (gamma1);
           m*.75 (sig2);
MODEL CONSTRAINT:
           NEW(ind*.45 dir*.4 arg11*.7 arg10*.25 arg01*.2 arg00*-
           .15 v1*1.6075 v0*1.3675 probit11*.5521 probit10*.1972
           probit01*.17103 probit00*-.1283 tie*.131 de*.129 pie*.119
           ortie*1.7788 orde*1.6808 orpie*1.614);
           dir=beta3*gamma0+beta2;
           ind=beta1*gamma1+beta3*gamma1;
           arg11=-mbeta0+beta2+(beta1+beta3)*(gamma0+gamma1);
           arg10=-mbeta0+beta2+(beta1+beta3)*gamma0;
           arg01=-mbeta0+beta1*(gamma0+gamma1);
           arg00=-mbeta0+beta1*gamma0;
           v1=(beta1+beta3)^2*sig2+1;
           v0=beta1^2*sig2+1;
           probit11=arg11/sqrt(v1);
           probit10=arg10/sqrt(v1);
           probit01=arg01/sqrt(v0);
           probit00=arg00/sqrt(v0);
           ! Phi function needed below:
           tie=phi(probit11)-phi(probit10);
           de=phi(probit10)-phi(probit00);
           pie=phi(probit01)-phi(probit00);
           ortie=(phi(probit11)/(1-phi(probit11)))/
           (phi(probit10)/(1-phi(probit10)));
           orde=(phi(probit10)/(1-phi(probit10)))/
           (phi(probit00)/(1-phi(probit00)));
           orpie=(phi(probit01)/(1-phi(probit01)))/
           (phi(probit00)/(1-phi(probit00)));

```

Table 30: Input for 1st rep step 2 bayes y on xm n=200

```

TITLE: 1st rep Binary Y, continuous M
DATA: FILE = 1stn200.dat;
VARIABLE:
      NAMES = y m x;
      USEVARIABLES = y m x xm;
      CATEGORICAL = y;
DEFINE: xm = x*m;
ANALYSIS:
      ESTIMATOR = BAYES;
      FBITER = 10000;
MODEL:
      [y$1*.5] (mbeta0);
      y ON x*.3 (beta2);
      y ON m*.7 (beta1);
      y ON xm*.2 (beta3);
      [m*.5] (gamma0);
      m ON x*.5 (gamma1);
      m*.75 (sig2);
MODEL CONSTRAINT:
      NEW(ind*.45 dir*.4 arg11*.7 arg10*.25 arg01*.2 arg00*-.15
v1*1.6075 v0*1.3675 probit11*.5521 probit10*.1972 probit01*.17103
probit00*-.1283 tie*.131 de*.129 pie*.119 ortie orde orpie);
      dir=beta3*gamma0+beta2;
      ind=beta1*gamma1+beta3*gamma1;
      arg11=-mbeta0+beta2+(beta1+beta3)*(gamma0+gamma1);
      arg10=-mbeta0+beta2+(beta1+beta3)*gamma0;
      arg01=-mbeta0+beta1*(gamma0+gamma1);
      arg00=-mbeta0+beta1*gamma0;
      v1=(beta1+beta3)^2*sig2+1;
      v0=beta1^2*sig2+1;
      probit11=arg11/sqrt(v1);
      probit10=arg10/sqrt(v1);
      probit01=arg01/sqrt(v0);
      probit00=arg00/sqrt(v0);
      ! Phi function needed below:
      tie=phi(probit11)-phi(probit10);
      de=phi(probit10)-phi(probit00);
      pie=phi(probit01)-phi(probit00);
      ortie=(phi(probit11)/(1-phi(probit11)))/
(phi(probit10)/(1-phi(probit10)));
      orde=(phi(probit10)/(1-phi(probit10)))/
(phi(probit00)/(1-phi(probit00)));
      orpie=(phi(probit01)/(1-phi(probit01)))/
(phi(probit00)/(1-phi(probit00)));
OUTPUT: TECH1 TECH8;
PLOT: TYPE = PLOT3;

```

Table 31: Input excerpts for juvcr on agg5 on tx agg1 tx-agg5 probit

```

USEVARIABLES = juvcr agg5 agg1 tx xm;
IDVARIABLE = prcid;
CATEGORIAL = juvcr;
USEOBSERVATION = gender EQ 1 AND (desgn11s EQ
1 OR desgn11s EQ 2 OR desgn11s EQ 3 OR desgn11s EQ
4);
DEFINE: IF(desgn11s EQ 4)THEN tx=1;
IF(desgn11s EQ 1 OR desgn11s EQ 2 OR desgn11s EQ
3)THEN tx=0;
agg1 = (sctaa11f-2.092)/1.0644;
agg5 = (sctaa15s-2.400)/1.100;
juvcr = juvadl;
xm = tx*agg5;
ANALYSIS: ESTIMATOR = MLR;
LINK = PROBIT;
INTEGRATION = MONTECARLO;
MODEL: [juvcr$1] (mbeta0);
juvcr ON tx (beta2)
agg5 (beta1)
xm (beta3)
agg1 (beta4);
[agg5] (gamma0);
agg5 ON tx (gamma1)
agg1 (gamma2);
agg5 (sig2);

```

Table 32: Input for juvcr on agg5 on tx agg1 tx-agg5 probit, continued

```

MODEL CONSTRAINT:
    NEW(ind dir arg11 arg10 arg00 v1 v0 probit11 probit10
    probit00 indirect direct orind ordir);
    dir=beta3*gamma0+beta2;
    ind=beta1*gamma1+beta3*gamma1;
    arg11=-mbeta0+beta2+beta4*0+(beta1+beta3)*(gamma0+
    gamma1+gamma2*0);
    arg10=-mbeta0+beta2+(beta1+beta3)*gamma0;
    arg00=-mbeta0+beta1*gamma0;
    v1=(beta1+beta3)^2*sig2+1;
    v0=beta1^2*sig2+1;
    probit11=arg11/sqrt(v1);
    probit10=arg10/sqrt(v1);
    probit00=arg00/sqrt(v0);
    ! Phi function needed below:
    indirect=phi(probit11)-phi(probit10);
    direct=phi(probit10)-phi(probit00);
    orind=(phi(probit11)/(1-phi(probit11)))/(phi(probit10)/(1-
    phi(probit10)));
    ordir=(phi(probit10)/(1-phi(probit10)))/(phi(probit00)/(1-
    phi(probit00)));
OUTPUT:  TECH1 TECH8 SAMPSTAT PATTERNS STANDARD-
        IZED;
PLOT:    TYPE = PLOT3;

```

Table 33: Input for juvcr on agg5 on tx agg1 tx-agg5 logit

```

USEVARIABLES = juvcr agg5 agg1 tx xm;
IDVARIABLE = prcid;
CATEGORICAL = juvcr;
USEOBSERVATION = gender EQ 1 AND (desgn1s EQ
1 OR desgn1s EQ 2 OR desgn1s EQ 3 OR desgn1s EQ
4);
DEFINE: IF(desgn1s EQ 4)THEN tx=1;
IF(desgn1s EQ 1 OR desgn1s EQ 2 OR desgn1s EQ
3)THEN tx=0;
agg1 = (sctaa11f-2.092)/1.0644;
agg5 = (sctaa15s-2.400)/1.100;
juvcr = juvadl;
xm = tx*agg5;
ANALYSIS: ESTIMATOR = MLR;
LINK = LOGIT;
INTEGRATION = MONTECARLO;
MODEL: [juvcr$1] (mbeta0);
juvcr ON tx (beta2)
agg5 (beta1)
xm (beta3)
agg1 (beta4);
[agg5] (gamma0);
agg5 ON tx (gamma1)
agg1 (gamma2);
agg5 (sig2);
MODEL CONSTRAINT:
NEW(ind*.45 dir*.4 oddsrat*.7);
dir=beta2+beta3*gamma0;
ind=beta1*gamma1+beta3*gamma1;
oddsrat=exp(ind);
OUTPUT: TECH1 TECH8 SAMPSTAT PATTERNS STANDARD-
IZED;
PLOT: TYPE = PLOT3;

```

Table 34: Input for m cont probit using maximum-likelihood

```

TITLE:   Clinical Trials data M (intent) treated as continuous
DATA:    FILE = 4cat m.dat;
VARIABLE:
          NAMES = intent tx ciguse w;
          USEVARIABLE = tx ciguse intent;
          CATEGORICAL = ciguse;
          FREQWEIGHT = w;
DEFINE:  intent = (intent-1.456)/0.8854;
ANALYSIS:
          ESTIMATOR = MLR;
          LINK = PROBIT;
MODEL:   [ciguse$1] (mbeta0);
          ciguse ON tx (beta2)
          intent (beta1)
          [intent] (gamma0);
          intent ON tx (gamma1);
          intent (sig2);
MODEL CONSTRAINT:
          NEW(ind dir arg11 arg10 arg00 v1 v0 probit11 probit10
          probit00 indirect direct orind ordir);
          dir=beta2;
          ind=beta1*gamma1;
          arg11=-mbeta0+beta2+beta1*(gamma0+gamma1);
          arg10=-mbeta0+beta2+beta1*gamma0;
          arg00=-mbeta0+beta1*gamma0;
          v1=beta1^2*sig2+1;
          v0=beta1^2*sig2+1;
          probit11=arg11/sqrt(v1);
          probit10=arg10/sqrt(v1);
          probit00=arg00/sqrt(v0);
          ! Phi function needed below:
          indirect=phi(probit11)-phi(probit10);
          direct=phi(probit10)-phi(probit00);
          orind=(phi(probit11)/(1-phi(probit11)))/(phi(probit10)/(1-
          phi(probit10)));
          ordir=(phi(probit10)/(1-phi(probit10)))/(phi(probit00)/(1-
          phi(probit00)));
OUTPUT:  TECH1 TECH8 SAMPSTAT PATTERNS STANDARD-
          IZED;
PLOT:    TYPE = PLOT3;

```

Table 35: Input for m* cont probit using weighted least-squares

```

TITLE:    Clinical Trials data using M* as mediator
DATA:    FILE = 4cat m.dat;
VARIABLE:
          NAMES = intent tx ciguse w;
          USEVARIABLE = tx ciguse intent;
          CATEGORICAL = ciguse intent;
          FREQWEIGHT = w;
ANALYSIS:
          ESTIMATOR = WLSMV;
MODEL:   [ciguse$1] (mbeta0);
          ciguse ON tx (beta2)
          intent (beta1)
          intent ON tx (gamma1);
MODEL CONSTRAINT:
          NEW(ind dir arg11 arg10 arg00 v1 v0 probit11 probit10
          probit00 indirect direct orind ordir);
          dir=beta2;
          ind=beta1*gamma1;
          arg11=-mbeta0+beta2+beta1*(0+gamma1); ! gamma0=0
          for m*
          arg10=-mbeta0+beta2+beta1*0;
          arg00=-mbeta0+beta1*0;
          v1=beta1^2*1+1; !sigma=1 for m* residual variance
          v0=beta1^2*1+1;
          probit11=arg11/sqrt(v1);
          probit10=arg10/sqrt(v1);
          probit00=arg00/sqrt(v0);
          ! Phi function needed below:
          indirect=phi(probit11)-phi(probit10);
          direct=phi(probit10)-phi(probit00);
          orind=(phi(probit11)/(1-phi(probit11)))/(phi(probit10)/(1-
          phi(probit10)));
          ordir=(phi(probit10)/(1-phi(probit10)))/(phi(probit00)/(1-
          phi(probit00)));
OUTPUT: TECH1 TECH8 SAMPSTAT PATTERNS STANDARD-
          IZED;
PLOT:    TYPE = PLOT3;

```

Table 36: Input for step 1 binary m binary y logit with xm interaction pearl ex
n=400 tie and pie

TITLE: Pearl (2011) artificial example, Prevention Science
 (Model Constraint notation below vs Pearl's: fm -> h, fy->g)
 Step 1: Generate and analyze the data using the same model

MONTECARLO:
 NAMES = y m x; ! x: tx/ctrl, m: mediator, y: outcome
 CATEGORICAL = y m;
 GENERATE = y(1 1) m(1 1); ! binary DVs, logit link
 CUTPOINTS = x(0); ! binary tx/ctrl dummy
 NOBSEVATION = 400;
 NREPS = 500;
 REPSAVE = ALL;
 SAVE = binbinlogitrep*.dat;

MODEL POPULATION:
 x@1; [x@0];
 [m\$1*.4055]; ! P(m=1 | x=0)=0.40
 m ON x*1.5041; ! P(m=1 | x=1)=0.75
 [y\$1*1.3863]; ! P(y=1 | x=0, m=0)=0.20.
 y ON x*.9808; ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
 beta1 | y ON m; ! defining the random slope
 [beta1*.539]; ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
 beta1 ON x*1.2528; ! interaction between x and m in their influence ON y.
 ! P(y=1 | x=1, m=1)=0.80
 beta1@0;

ANALYSIS:
 TYPE = RANDOM;
 ESTIMATOR = ML;
 LINK = LOGIT;

MODEL:
 [m\$1*.4055] (Th0); ! Threshold=-Logit. P(m=1 | x=0)=0.40
 m ON x*1.5041 (Lh1); ! P(m=1 | x=1)=0.75
 [y\$1*1.3863] (Tg00); ! Threshold=-Logit. P(y=1 | x=0,m=0)=0.20
 y ON x*.9808 (Lg10); ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
 beta1 | y ON m; ! defining the random slope
 [beta1*.539]; ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
 beta1 ON x*1.2528; ! interaction between x and m in their influence ON y.
 ! P(y=1 | x=1, m=1)=0.80
 beta1@0;

Table 37: Input for step 1 binary m binary y logit with xm interaction pearl ex
n=400 tie and pie, continued

MODEL CONSTRAINT:

```

NEW(fm0*.4 fm1*.75 fy00*.2 fy10*.4 fy01*.3 fy11*.8
de*.32 pie*.035 tie*.14 te*.46 iete*.07 dete*.696
compdete*.304 tietete*.304);
fm0=1/(1+exp(Th0));
fm1=1/(1+exp(Th0-Lh1));
fy00=1/(1+exp(Tg00));
fy10=1/(1+exp(Tg00-Lg10));
fy01=1/(1+exp(Tg00-Lg01));
fy11=1/(1+exp(Tg00-Lg10-Lg01-Lg11));
de=(fy10-fy00)*(1-fm0)+(fy11-fy01)*fm0;
tie=(fy11-fy10)*(fm1-fm0);
pie=(fy01-fy00)*(fm1-fm0);
te=fy11*fm1+fy10*(1-fm1)-(fy01*fm0+fy00*(1-fm0));
iete=pie/te;
dete=de/te;
compdete=1-dete/te;
tietete=tie/te;

```

Table 38: Input for step 2 define xm binary m binary y logit with xm interaction
 pearl ex n=400 tie and pie

```

TITLE: Pearl (2011) artificial example, Prevention Science (Model Constraint
      notation below vs Pearl's: fm -> h, fy->g)
      Step 2: Analyze the data using an interaction variable
DATA:
      FILE = binbinlogitreplist.dat;
      TYPE = MONTECARLO;
VARIABLE:
      NAMES = y m x; ! x: tx/ctrl, m: mediator, y: outcome
      CATEGORICAL = y m;
      USEVARIABLES = y m x xm;
DEFINE:
      xm = x*m; ! create the interaction variable
ANALYSIS:
      ESTIMATOR = ML;
      LINK = LOGIT;
MODEL:
      [m$1*.4055] (Th0); ! Threshold=-Logit. P(m=1 | x=0)=0.40
      m ON x*1.5041 (Lh1); ! P(m=1 | x=1)=0.75
      [y$1*1.3863] (Tg00); ! Threshold=-Logit. P(y=1 | x=0,m=0)=0.20
      y ON x*.9808 (Lg10); ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
      y ON m*.539 (Lg01); ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
      y ON xm*1.2528 (Lg11); ! interaction between x and m
      ! in their influence ON y. P(y=1 | x=1,m=1)=0.80
MODEL CONSTRAINT:
      NEW(fm0*.4 fm1*.75 fy00*.2 fy10*.4 fy01*.3 fy11*.8
      de*.32 pie*.035 tie*.14 te*.46 iete*.07 dete*.696
      compdete*.304 tietete*.304);
      fm0=1/(1+exp(Th0));
      fm1=1/(1+exp(Th0-Lh1));
      fy00=1/(1+exp(Tg00));
      fy10=1/(1+exp(Tg00-Lg10));
      fy01=1/(1+exp(Tg00-Lg01));
      fy11=1/(1+exp(Tg00-Lg10-Lg01-Lg11));
      de=(fy10-fy00)*(1-fm0)+(fy11-fy01)*fm0;
      tie=(fy11-fy10)*(fm1-fm0);
      pie=(fy01-fy00)*(fm1-fm0);
      te=fy11*fm1+fy10*(1-fm1)-(fy01*fm0+fy00*(1-fm0));
      iete=pie/te;
      dete=de/te;
      compdete=1-de/te;
      tietete=tie/te;

```

Table 39: Input for step 1 binary m binary y probit with xm interaction pearl ex n=400

TITLE: Pearl (2011) artificial example, Prevention Science Probit link (Pearl->Mplus notation: h->fm, g->fy)

MONTECARLO:

NAMES = y m x; !x: tx/ctrl, m: mediator, y: outcome
 CATEGORICAL = y m;
 GENERATE = y(1 p) m(1 p); ! binary DVs, probit link
 CUTPOINTS = x(0); ! binary tx/ctrl dummy
 NOBSEVATION = 400;
 NREPS = 500;
 REPSAVE = ALL;
 SAVE = binbinprobrep*.dat;

MODEL POPULATION:

x@1; [x@0];
 [m\$1*.254]; ! P(m=1 | x=0)=0.40
 m ON x*.929; ! P(m=1 | x=1)=0.75
 [y\$1*.84]; ! P(y=1 | x=0,m=0)=0.20
 y ON x*.586; ! direct effect of x ON y. P(y=1 |x=1,m=0)=0.40
 beta1 | y ON m;
 [beta1*.315]; ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
 beta1 ON x*.779; ! interaction between x AND m in their influence ON y
 !P(y=1 | x=1,m=1)=0.80
 beta1@0;

ANALYSIS:

TYPE = RANDOM;
 ESTIMATOR = ML;
 LINK = PROBIT;

MODEL:

[m\$1*.254] (fm0); ! Negative intercept. P(m=1 | x=0)=0.40
 m ON x*.929 (fm1); ! P(m=1 | x=1)=0.75
 [y\$1*.84] (fy00); ! Negative intercept. P(y=1 | x=0,m=0)=0.20
 y ON x*.586 (fy10); ! direct effect of x ON y. P(y=1 |x=1,m=0)=0.40
 beta1 | y ON m;
 [beta1*.315] (fy01); ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
 beta1 ON x*.779 (fy11); ! interaction between x AND m in their influence
 ON y
 !P(y=1 | x=1,m=1)=0.80
 beta1@0;

Table 40: Input for step 1 binary m binary y probit with xm interaction pearl ex n=400, continued

MODEL CONSTRAINT:

```

NEW(de*.32 tie*.14 pie*.035 te*.46 tiete*.304 piete*.07 dete*.696
compdete*.304 pfm0*.4 pfm1*.75 pfy00*.2 pfy10*.4 pfy01*.3 pfy11*.8);
pfm0=phi(-fm0);
pfm1=phi(-fm0+fm1);
pfy00=phi(-fy00);
pfy10=phi(-fy00+fy10);
pfy01=phi(-fy00+fy01);
pfy11=phi(-fy00+fy10+fy01+fy11);
de=(pfy10-pfy00)*(1-pfm0)+(pfy11-pfy01)*pfm0;
tie=(pfy11-pfy10)*(pfm1-pfm0);
pie=(pfy01-pfy00)*(pfm1-pfm0);
te=pfy11*pfm1+pfy10*(1-pfm1) -(pfy01*pfm0+pfy00*(1-pfm0));
tiete=tie/te;
piete=pie/te;
dete=de/te;
compdete=1-de/te;

```

Table 41: Input for step 2 ml define xm binary m binary y probit with xm interaction pearl ex n=400

```

TITLE: Pearl (2011) artificial example, Prevention Science Probit. Step 2
DATA: FILE = binbinprobreplist.dat;
      TYPE = MONTECARLO;
VARIABLE:
      NAMES = y m x; lx: tx/ctrl, m: mediator, y: outcome
      CATEGORICAL = y m;
      USEVARIABLES = y m x xm;
DEFINE:
      xm = x*m;
ANALYSIS:
      ESTIMATOR = ML;
      LINK = PROBIT;
MODEL:
      [m$1*.254] (fm0); ! Negative intercept. P(m=1 | x=0)=0.40
      m on x*.929 (fm1); ! P(m=1 | x=1)=0.75
      [y$1*.84] (fy00); ! Negative intercept. P(y=1 | x=0,m=0)=0.20
      y ON x*.586 (fy10); ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
      y ON m*.315 (fy01); ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
      y ON xm*.779 (fy11); ! interaction between x AND m in their influence
      ! ON y. P(y=1 | x=1,m=1)=0.80
MODEL CONSTRAINT:
      NEW(de*.32 tie*.14 pie*.035 te*.46 tiete*.304 piete*.07 dete*.696
      compdete*.304 pfm0*.4 pfm1*.75 pfy00*.2 pfy10*.4 pfy01*.3 pfy11*.8);
      pfm0=phi(-fm0);
      pfm1=phi(-fm0+fm1);
      pfy00=phi(-fy00);
      pfy10=phi(-fy00+fy10);
      pfy01=phi(-fy00+fy01);
      pfy11=phi(-fy00+fy10+fy01+fy11);
      de=(pfy10-pfy00)*(1-pfm0)+(pfy11-pfy01)*pfm0;
      tie=(fy11-pfy10)*(pfm1-pfm0);
      pie=(pfy01-pfy00)*(pfm1-pfm0);
      te=pfy11*pfm1+pfy10*(1-pfm1) - (pfy01*pfm0+pfy00*(1-pfm0));
      tiete=tie/te;
      piete=pie/te;
      dete=de/te;
      compdete=1-de/te;

```

Table 42: Input for step 2 bayes define xm binary m binary probit with xm interaction pearl ex n=400 10k

```

TITLE:
    Pearl (2011) artificial example, Prevention Science
    Step 2. Bayes analysis
DATA:
    FILE = binbinprobreplist.dat;
    TYPE = MONTECARLO;
VARIABLE:
    NAMES = y m x; !x: tx/ctrl, m: mediator, y: outcome
    CATEGORICAL = y m;
    USEVARIABLES = y m x xm;
DEFINE:
    xm = x*m;
ANALYSIS:
    ESTIMATOR = BAYES;
    FBITER = 10000;
    MEDIATOR = OBSERVED;
MODEL:
    [m$1*.254] (fm0); ! Negative intercept. P(m=1 | x=0)=0.40
    m ON x*.929 (fm1); ! P(m=1 | x=1)=0.75
    [y$1*.84] (fy00); ! Negative intercept. P(y=1 | x=0,m=0)=0.20
    y ON x*.586 (fy10); ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
    y ON m*.315 (fy01); ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
    y ON xm*.779 (fy11); ! interaction between x and m in their
    ! influence ON y. P(y=1 | x=1,m=1)=0.80

```

Table 43: Input for step 2 bayes define xm binary m binary probit with xm interaction pearl ex n=400 10k, continued

MODEL CONSTRAINT:

```

NEW(de*.32 tie*.14 pie*.035 te*.46 tiete*.304 piete*.07
dete*.696 compdete*.304 orde*4.0303 ortie*1.8333 pfm0 pfm1
pfy00 pfy10 pfy01 pfy11 numde dende numtie dentie);
pfm0=phi(-fm0);
pfm1=phi(-fm0+fm1);
pfy00=phi(-fy00);
pfy10=phi(-fy00+fy10);
pfy01=phi(-fy00+fy01);
pfy11=phi(-fy00+fy10+fy01+fy11);
de=(pfy10-pfy00)*(1-pfm0)+(pfy11-pfy01)*pfm0;
tie=(pfy11-pfy10)*(pfm1-pfm0);
pie=(pfy01-pfy00)*(pfm1-pfm0);
te=pfy11*pfm1+pfy10*(1-pfm1)-(pfy01*pfm0+pfy00*(1-
pfm0));
tiete=tie/te;
piete=pie/te;
dete=de/te;
compdete=1-de/te;
numde=pfy10*(1-pfm0)+pfy11*pfm0;
dende=pfy00*(1-pfm0)+pfy01*pfm0;
orde=(numde/(1-numde))/(dende/(1-dende));
numtie=pfy10*(1-pfm1)+pfy11*pfm1;
dentie=pfy10*(1-pfm0)+pfy11*pfm0;
ortie=(numtie/(1-numtie))/(dentie/(1-dentie));

```

Table 44: Input for Bayes analysis of n=200 data drawn on the Pearl example

```

TITLE: Pearl (2011) artificial example for n=200 probit.
DATA: FILE = n200.dat;
VARIABLE:
    NAMES = y m x w; ! x: tx/ctrl, m: mediator, y: outcome
    FREQWEIGHT = w;
    CATEGORICAL = y m;
    USEVARIABLES = y m x xm;
DEFINE:
    xm = x*m;
ANALYSIS:
    ESTIMATOR = BAYES;
    FBITER = 10000;
    MEDIATOR = OBSERVED;
MODEL:
    [m$1*.254] (fm0); ! Negative intercept. P(m=1 | x=0)=0.40
    m ON x*.929 (fm1); ! P(m=1 | x=1)=0.75
    [y$1*.84] (fy00); ! Negative intercept. P(y=1 | x=0,m=0)=0.20
    y ON x*.586 (fy10); ! direct effect of x ON y. P(y=1 | x=1,m=0)=0.40
    y ON m*.315 (fy01); ! main effect of m ON y. P(y=1 | x=0,m=1)=0.30
    y ON xm*.779 (fy11); ! interaction between x and m in their influence
    ! ON y P(y=1 | x=1,m=1)=0.80

```

Table 45: Input for Bayes analysis of n=200 data drawn on the Pearl example, continued

```

MODEL CONSTRAINT:
  NEW(de*.32 tie*.14 pie*.035 te*.46 tiete*.304 piete*.07
dete*.696 compdete*.304 pfm0*.4 pfm1*.75 pfy00*.2 pfy10*.4
pfy01*.3 pfy11*.8 numde*.56 dende*.24 orde*4.0303 numind*.7
denind*.56 orind*1.8333);
  pfm0=phi(-fm0);
  pfm1=phi(-fm0+fm1);
  pfy00=phi(-fy00);
  pfy10=phi(-fy00+fy10);
  pfy01=phi(-fy00+fy01);
  pfy11=phi(-fy00+fy10+fy01+fy11);
  de=(pfy10-pfy00)*(1-pfm0)+(pfy11-pfy01)*pfm0;
  tie=(pfy11-pfy10)*(pfm1-pfm0);
  pie=(pfy01-pfy00)*(pfm1-pfm0);
  te=pfy11*pfm1+pfy10*(1-pfm1)-(pfy01*pfm0+pfy00*(1-
pfm0));
  tiete=tie/te;
  piete=pie/te;
  dete=de/te;
  compdete=1-de/te;
  numde=pfy10*(1-pfm0)+pfy11*pfm0;
  dende=pfy00*(1-pfm0)+pfy01*pfm0;
  orde=(numde/(1-numde))/(dende/(1-dende));
  numind=pfy10*(1-pfm1)+pfy11*pfm1;
  denind=pfy10*(1-pfm0)+pfy11*pfm0;
  orind=(numind/(1-numind))/(denind/(1-denind));
OUTPUT:
  TECH1 TECH8;
PLOT: TYPE = PLOT3;

```

Table 46: Input for step 1 y on xm n=800

TITLE: Nominal M, Continuous Y
Using a latent class variable to represent M
Simulating x-m interaction effect ON y by class-varying y
ON x
Step 1: Saving the data for external Monte Carlo analysis

MONTECARLO:
NAMES = y x;
GENCLASSES = c(3);
CLASSES = c(3);
NOBSERVATION = 800;
NREPS = 500;
REPSAVE = ALL;
SAVE = n800xmrep*.dat;
CUTPOINTS = x(0);

MODEL POPULATION:
%OVERALL%
x@1;
[c#1*-1];
[c#2*-.5];
c#1 ON x*.7;
c#2 ON x*.3;
y ON x*0;
y*.75;
%c#1%
[y*-2];
y ON x*-.5;
%c#2%
[y*0];
y ON x*-.3;
%c#3%
[y*2];
y ON x*-.2;

Table 47: Input for step 1 y on xm n=800, continued

```

ANALYSIS: TYPE = MIXTURE;
           ESTIMATOR = ML;
MODEL:    %OVERALL%
          [c#1*-1];
          [c#2*-.5];
          c#1 ON x*.7;
          c#2 ON x*.3;
          y ON x*0;
          y*.75;
          %c#1%
          [y*-2];
          y ON x*-.5;
          %c#2%
          [y*0];
          y ON x*-.3;
          %c#3%
          [y*2];
          y ON x*-.2;

```

Table 48: Input for step 2 y on xm knownclass

```

TITLE:      Nominal M, Continuous Y
            Using a latent class variable to represent M
            Simulating x-m interaction effect ON y by class-varying y
            ON x
            Step 2: External Monte Carlo Analysis
DATA:      FILE = n800xmreplist.dat;
            TYPE = MONTECARLO;
VARIABLE:  NAMES = y x m;
            USEVARIABLES = y x;
            CLASSES = c(3);
            KNOWNCLASS = c(m=1 m=2 m=3);
ANALYSIS:  TYPE = MIXTURE;
            ESTIMATOR = ML;
MODEL:     %OVERALL%
            [c#1*-1] (gamma01);
            [c#2*-.5] (gamma02);
            c#1 ON x*.7 (gamma11);
            c#2 ON x*.3 (gamma12);
            y ON x*0;
            y*.75;
            %c#1%
            [y*-2] (beta01);
            y ON x*-.5 (beta11);
            %c#2%
            [y*0] (beta02);
            y ON x*-.3 (beta12);
            %c#3%
            [y*2] (beta03);
            y ON x*-.2 (beta13);

```

Table 49: Input for step 2 y on xm knownclass, continued

MODEL CONSTRAINT:

```

NEW(denom0*1.9744      denom1*2.5595      p10*.1863
p11*.2894  p20*.3072  p21*.3199  p30*.5065  p31*.3907
term11*-.1162 term10*.3538 term01*.2026 term00*.6404
de*-.2866 tie*-.47 total*-.7566 pie*-.4378);
! index is x' for multinomial denominator
denom0=exp(gamma01)+exp(gamma02)+1;
denom1=exp(gamma01+gamma11)+exp(gamma02+gamma12)+1;
! first index is class, second x' for probabilities
p10=exp(gamma01)/denom0;
p11=exp(gamma01+gamma11)/denom1;
p20=exp(gamma02)/denom0;
p21=exp(gamma02+gamma12)/denom1;
p30=1/denom0;
p31=1/denom1;
! first index is x, second x', summing over class
term11=(beta01+beta11)*p11+(beta02+beta12)*p21
+(beta03+beta13)*p31;
term10=(beta01+beta11)*p10+(beta02+beta12)*p20
+(beta03+beta13)*p30;
term01=(beta01)*p11+(beta02)*p21+(beta03)*p31;
term00=(beta01)*p10+(beta02)*p20+(beta03)*p30;
de=term10-term00;
tie=term11-term10;
total=term11-term00;
pie=term01-term00;

```

Table 50: Input for hypothetical pollution data with a nominal mediator and a binary outcome

```

TITLE:      Nominal M, Binary Y
            Using a latent class variable to represent M
DATA:      FILE = nombin.dat;
VARIABLE:  NAMES = x m y w;
            FREQWEIGHT = w;
            USEVARIABLES = y x;
            CATEGORICAL = y;
            CLASSES = c(3);
            KNOWNCLASS = c(m=1 m=2 m=3);
ANALYSIS:  TYPE = MIXTURE;
            ESTIMATOR = ML;
MODEL:     %OVERALL%
            [c#1] (gamma01);
            [c#2] (gamma02);
            c#1 ON x (gamma11);
            c#2 ON x (gamma12);
            y ON x;
            %c#1%
            [y$1] (beta01);
            y ON x (beta11);
            %c#2%
            [y$1] (beta02);
            y ON x (beta12);
            %c#3%
            [y$1] (beta03);
            y ON x (beta13);

```

Table 51: Input for hypothetical pollution data with a nominal mediator and a binary outcome, continued

MODEL CONSTRAINT:

```

NEW(denom0 denom1 p10 p11 p20 p21 p30 p31 term11
term10 term01 term00 de tie total pie orde ortie orpie);
! index is x' for multinomial denominator
denom0=exp(gamma01)+exp(gamma02)+1;
denom1=exp(gamma01+gamma11)+exp(gamma02+gamma12)+1;
! first index is class, second x' for probabilities
p10=exp(gamma01)/denom0;
p11=exp(gamma01+gamma11)/denom1;
p20=exp(gamma02)/denom0;
p21=exp(gamma02+gamma12)/denom1;
p30=1/denom0;
p31=1/denom1;
! first index is x, second x', summing over class
term11=(1/(1+exp(beta01-beta11)))*p11+(1/(1+exp(beta02-
beta12)))*p21 +(1/(1+exp(beta03-beta13)))*p31;
term10=(1/(1+exp(beta01-beta11)))*p10+(1/(1+exp(beta02-
beta12)))*p20 +(1/(1+exp(beta03-beta13)))*p30;
term01=(1/(1+exp(beta01)))*p11+(1/(1+exp(beta02)))*p21
+(1/(1+exp(beta03)))*p31;
term00=(1/(1+exp(beta01)))*p10+(1/(1+exp(beta02)))*p20
+(1/(1+exp(beta03)))*p30;
de=term10-term00;
tie=term11-term10;
total=term11-term00;
pie=term01-term00;
orde=(term10/(1-term10))/(term00/(1-term00));
ortie=(term11/(1-term11))/(term10/(1-term10));
orpie=(term01/(1-term01))/(term00/(1-term00));

```

OUTPUT: TECH1 TECH8;

Table 52: Input for step 1 count y on xm

TITLE: Simulating x-m interaction effect on y using a random slope
 Count Y, continuous M
 Step 1: saving the data for external Monte Carlo analysis

MONTECARLO:

```

NAMES = y m x;
GENERATE = y(p);
COUNT = y(p);
NOBSERVATION = 400;
NREPS = 500;
REPSAVE = ALL;
SAVE = xmrep*.dat;
CUTPOINTS = x(0);

```

MODEL POPULATION:

```

x@1;
[y*-.7]; ! this log rate gives a rate of about 0.5
y ON x*.3;
beta1 | y ON m;
beta1 ON x*.2;
[beta1*.4];
beta1@0;
[m*.5];
m ON x*.5;
m*.75;

```

ANALYSIS: TYPE = RANDOM;
ESTIMATOR = ML;

MODEL: [y*-.7] (beta0);
y ON x*.3 (beta2);
beta1 | y ON m;
beta1 ON x*.2 (beta3);
[beta1*.4] (beta1);
beta1@0;
[m*.5] (gamma0);
m ON x*.5 (gamma1);
m*.75 (sig);

Table 53: Input for step 1 count y on xm, continued

MODEL CONSTRAINT:

```

NEW(ind*.45 dir*.4 ey1 ey0 mum1 mum0 ay1 ay0 bym11
by m10 by m01 by m00 eym11 eym10 eym01 eym00 tie de
total pie);
ind=beta1*gamma1+beta3*gamma1;
dir=beta3*gamma0+beta2;
ey1=exp(beta0+beta2);
ey0=exp(beta0);
mum1=gamma0+gamma1;
mum0=gamma0;
ay1=2*sig*(beta1+beta3);
ay0=2*sig*beta1;
bym11=(ay1/mum1+2)/2;
bym10=(ay1/mum0+2)/2;
bym01=(ay0/mum1+2)/2;
bym00=(ay0/mum0+2)/2;
eym11=exp((bym11*bym11-1)*mum1*mum1/(2*sig));
eym10=exp((bym10*bym10-1)*mum0*mum0/(2*sig));
eym01=exp((bym01*bym01-1)*mum1*mum1/(2*sig));
eym00=exp((bym00*bym00-1)*mum0*mum0/(2*sig));
tie=ey1*eym11-ey1*eym10;
de=ey1*eym10-ey0*eym00;
total=ey1*eym11-ey0*eym00;
pie=ey0*eym01-ey0*eym00;

```

Table 54: Input for step 2 count y on xm

```

TITLE:      Simulating x-m interaction effect on y using a random slope
            Binary Y, continuous M
            Step 2: External Monte Carlo analysis
DATA:      FILE = xmreplist.dat;
            TYPE = MONTECARLO;
VARIABLE:  NAMES = y m x;
            USEVARIABLES = y m x xm;
            COUNT = y(p);
DEFINE:    xm = x*m;
ANALYSIS:  ESTIMATOR = ML;
MODEL:     [y*-.7] (beta0);
            y ON x*.3 (beta2);
            y ON m*.4 (beta1);
            y ON xm*.2 (beta3);
            [m*.5] (gamma0);
            m on x*.5 (gamma1);
            m*.75 (sig);
MODEL CONSTRAINT:
            NEW(ind*.45 dir*.4 ey1*.67032 ey0*.49658 mum1*1
            mum0*.5 ay1*.9 ay0*.6 bym11*1.45 bym10*1.9 bym01*1.3
            bym00*1.6 eym11*2.0855 eym10*1.5450 eym01*1.5841
            eym00*1.2969 tie*.3361 de*.3916 total*.7539 pie*.1426);
            ind=beta1*gamma1+beta3*gamma1;
            dir=beta3*gamma0+beta2;
            ey1=exp(beta0+beta2);
            ey0=exp(beta0);
            mum1=gamma0+gamma1;
            mum0=gamma0;
            ay1=2*sig*(beta1+beta3);
            ay0=2*sig*beta1;
            bym11=(ay1/mum1+2)/2;
            bym10=(ay1/mum0+2)/2;
            bym01=(ay0/mum1+2)/2;
            bym00=(ay0/mum0+2)/2;
            eym11=exp((bym11*bym11-1)*mum1*mum1/(2*sig));
            eym10=exp((bym10*bym10-1)*mum0*mum0/(2*sig));
            eym01=exp((bym01*bym01-1)*mum1*mum1/(2*sig));
            eym00=exp((bym00*bym00-1)*mum0*mum0/(2*sig));
            tie=ey1*eym11-ey1*eym10;
            de=ey1*eym10-ey0*eym00;
            total=ey1*eym11-ey0*eym00;
            pie=ey0*eym01-ey0*eym00;

```

Table 55: Input for rho=0 run: replicating regular mediation analysis

```

TITLE:      Generate as M on X, Y on M and X
            Analyze as M on X, Y on X
            Using Imai formulas with rho=0,
            so same as regular mediation run
MONTECARLO:
            NAMES = y m x;
            NOBSERVATION = 400;
            NREPS= 500;
            CUTPOINTS = x(0);
MODEL POPULATION:
            x@1;
            [y*1]; ! beta0
            y ON m*.5 ! beta1
            x*.4; ! beta2
            [m*2]; ! gamma0
            m ON x*.5; ! gamma1
            y*.5; ! sig1
            m*1; ! sig2
ANALYSIS:
MODEL:      [y] (kappa0);
            y ON x (kappa1);
            [m*2] (gamma0);
            m ON x*.5 (gamma1);
            y*.75 (sig);
            m*1 (sig2);
            y WITH m*.5 (cov);
MODEL CONSTRAINT:
            NEW(rhocurl*.5774 beta1*.5 beta2*.4 beta0*1 sig1*.5
            ind*.25 de*.4);
            rhocurl=cov/(sqrt(sig)*sqrt(sig2));
            ! derive the data-generating model parameters
            ! from the analysis model parameters:
            beta1=rhocurl*sqrt(sig)/sqrt(sig2);
            beta2=kappa1-beta1*gamma1;
            beta0=kappa0-beta1*gamma0;
            sig1=sig-beta1*beta1*sig2;
            ! express the effects in terms of the data-generating
            ! parameters:
            ind=beta1*gamma1;
            de=beta2;

```

Table 56: Input for true corr=0.25, rho=0.25

TITLE: Generate as M on X, Y on M and X
adding a residual correlation of rho=0.25
Analyze as M on X, Y on X
Using Imai formulas for sensitivity analysis

MONTECARLO:
 NAMES = y m x;
 NOBSERVATION = 400;
 NREPS = 500;
 CUTPOINTS = x(0);

MODEL POPULATION:
 x@1;
 [y*1]; ! beta0
 y ON m*.5 ! beta1
 x*.4; ! beta2
 [m*2]; ! gamma0
 m ON x*.5; ! gamma1
 y*.5; ! sig1
 m*1; ! sig2
 y WITH m*.1768; ! gives res corr 0.25

ANALYSIS:
MODEL: [y] (kappa0);
 y ON x (kappa1);
 [m*2] (gamma0);
 m ON x*.5 (gamma1);
 y*1.1036 (sig);
 m*1 (sig2);
 y WITH m*.8536 (cov);

MODEL CONSTRAINT:
 NEW(rho*.25 rhocurl*.8125 beta1*.5 beta2*.4 beta0*1
sig1*.7071 ind*.25 de*.4);
 rhocurl=cov/(sqrt(sig)*sqrt(sig2));
 rho=.25;
 beta1=(sqrt(sig)/sqrt(sig2))*
 (rhocurl-rho*sqrt((1-rhocurl*rhocurl)/(1-rho*rho)));
 beta2=kappa1-beta1*gamma1;
 beta0=kappa0-beta1*gamma0;
 sig1=(rhocurl*sqrt(sig)-beta1*sqrt(sig2))/0.5;
 ind=beta1*gamma1;
 de=beta2;

Table 57: Input excerpts for head circumference analysis with $\rho=0$ corresponding to regular mediation analysis

```

MODEL:      [hcirc36] (kappa0);
            hcirc36 ON alccig (kappa1)
            gender eth;
            [hcirc0] (gamma0);
            hcirc0 ON alccig*1 (gamma1)
            gender eth;
            hcirc36*1 (sig);
            hcirc0*1 (sig2);
            hcirc36 WITH hcirc0 (cov);
MODEL CONSTRAINT:
            NEW(rho rhocurl beta1 beta2 beta0 sig1 indirect direct);
            rhocurl=cov/(sqrt(sig)*sqrt(sig2));
            rho=0;
            beta1=(sqrt(sig)/sqrt(sig2))*
            (rhocurl-rho*sqrt((1-rhocurl*rhocurl)/(1-rho*rho)));
            beta2=kappa1-beta1*gamma1;
            beta0=kappa0-beta1*gamma0;
            sig1=(rhocurl*sqrt(sig)-beta1*sqrt(sig2))/0.5;
            indirect=beta1*gamma1;
            direct=beta2;
OUTPUT:     SAMPSTAT STANDARDIZED;
PLOT:       TYPE = PLOT3;

```
